

SETLIBL A Simple But Useful Utility



Graham Cromar

By Graham L. Cromar

Sometimes a simple utility can be the best. The command presented here is by no means sophisticated*. It may not even be considered instructive for that matter. However, it is very useful if you frequently need to change the user portion of your library list.

Let's say, for example, you have two versions of the same source in different libraries. This keeps the modifications for a certain site (e.g., a U.S. plant) separate from the version run at another site (e.g. Canada). Do this to accommodate modifications to several packages and soon the library list on the Canadian site will be quite different than in the U.S. Now add modifications for Europe, subsets of users using different modifications for specific clients ... you get the picture.

If you have to test software modifications for these environments you will find yourself doing a lot of typing in the IBM supplied EDTLIBL command. This is both time-consuming and prone to error. Besides, if you find yourself doing repetitive tasks on a computer, you're not using it right.

Enter the SETLIBL command. It accepts the library/filename/member of a multi-member file, which holds a pre-defined library list in each member. A couple of short CL programs replace your library list with the list in the specified member. A default member holds your initial library list so you can change back easily.

The Specifics

1. Create a multi-member file named LIBFIL in QGPL using the DDS below and the following command:

```
CRTPF FILE(QGPL/LIBFIL)
SRCFILE(*LIBL/QDDSSRC)
MAXMBRS(*NOMAX)
```

2. Compile programs SETLIBC1 and SETLIBC2 as follows (two programs are needed because it is not possible to do the override before the file declaration in a single program):

```
CRTCLPGM PGM(QGPL/
SETLIBC1)
```

```
CRTCLPGM PGM(QGPL/
SETLIBC2)
```

3. Create the SETLIBL command using the command source below:

```
CRTCMD CMD(QGPL/
SETLIBL) PGM(QGPL/
SETLIBC1) SRCFILE(QGPL/
QCMSDRC)
```

4. Add members to LIBFIL for each library list you want to save. You should at least add a member named "DEFAULT" at this time:

```
ADDPFM FILE(QGPL/
LIBFIL) MBR(DEFAULT)
TEXT('Default library list')
```

5. Use your favorite file editor (e.g. DBU) to add your default library list to the member. Add other members as needed.

These steps will place the command, programs and multi-member file in library QGPL. If you wish to use a different file (e.g. if you have several programmers that want to define their own custom library lists in their own file) the SETLIBL command will accommodate this. But, note that the default values point to the file in QGPL and the program will not behave as you want if the default file is not present and the parameter values are not supplied.

If you have a different library for common utilities and you wish to place the objects there instead of in QGPL,

I would suggest changing the default values in the command source to reflect this library. You will need to substitute this library for QGPL in the above instructions as well.

Running the command

If you are using the default filename as defined in the command source (i.e. LIBFIL) this command is extremely easy to run. Simply enter "SETLIBL" and hit <enter> to change your library list to the default list you saved. To change to any other list, enter "SETLIBL MEMBERNAME" where MEMBERNAME is the name of the member in LIBFIL that contains the library list you want. **Hint:** Use meaningful member names when designing your lists! If you are directing the command to use a file other than the default, you will need to enter the library, filename and member. You can prompt the command to view the required parameters.

Other uses

The SETLIBL command allows you to soft-code user library lists. These can then be changed without recompiling CL programs.



Graham Cromar can be reached via email at: gcromar@source-code-technologies.com

*For a more sophisticated solution to this problem see: iSeriesNEWS July, 1996 ARTICLE ID:1538 by Peter Clifford and Ernie Malaga.

SETLIBL - Source Code

```

A          R LIBFILR          TEXT('SAVED LIBRARY LISTS')
A          LIBNAM            10A          TEXT('LIBRARY NAME')

/*****
/* SETLIBC1: Graham Cromar; March 10, 2003 */
/*
/* Function: Set up file override and pass control to SETLIBC2.
/*
/* If no parameters are overridden, the command will
/* restore the library list saved in the member
/* "QGPL/LIBFIL/DEFAULT".
/*
*****/
PGM          PARM(&MBR &FIL &LIB)

DCL          &MBR          *CHAR 10
DCL          &FIL          *CHAR 10
DCL          &LIB          *CHAR 10
DCL          &MSG          *CHAR 80

CHKOBJ      OBJ(&LIB/&FIL) OBJTYPE(*FILE) MBR(&MBR)
MONMSG     MSGID(CPF0000) EXEC(DO)
CHGVAR     VAR(&MSG) VALUE('Aborted: Member `*bcat &mbr +
                          `*bcat `in file `*bcat &fil *bcat +
                          `not found in Library `*bcat &lib)

SNDPGMMSG  MSG(&MSG)
GOTO      CMDLBL(ABORT)
ENDDO

OVRDBF     FILE(LIBFIL) TOFILE(&LIB/&FIL) MBR(&MBR)
MONMSG     MSGID(CPF0000) EXEC(DO)
CHGVAR     VAR(&MSG) VALUE('Aborted: Could not override +
                          to specified file:`*bcat &lib *tcat `/' +
                          *cat &fil *tcat `/' *cat &mbr)

SNDPGMMSG  MSG(&MSG)
GOTO      CMDLBL(ABORT)
ENDDO

CALL       PGM(SETLIBC2)

DLTOVR     LIBFIL
ABORT:     ENDPGM

/*****
/* SETLIBC2: Graham Cromar; March 10, 2003 */
/*
/* Function: Set up a custom library list based on a previously
/* stored list of libraries in a multi-member file.
/*
*****/
PGM

DCLF       FILE(LIBFIL)

CHGLIBL    LIBL(*NONE)

DO
LOOP1:     RCVF          RCDfmt(*FILE)
MONMSG     MSGID(CPF0864) EXEC(GOTO CMDLBL(EOF))
ADDLIBLE   LIB(&LIBNAM) POSITION(*LAST)
MONMSG     MSGID(CPF0000)
GOTO LOOP1
RETURN

ENDPGM

/*****
/* SETLIBL: Graham Cromar; March 10, 2003 */
/*
/* Function: Prompt for lib/fil/mbr containing pre-stored
/* library list.
/*
*****/
CMD        PROMPT('Set Library List')

PARM       KWD(MBR) TYPE(*NAME) LEN(10) DFT(DEFAULT) +
          PROMPT('Member')

PARM       KWD(FIL) TYPE(*NAME) LEN(10) DFT(LIBFIL) +
          PROMPT('File')

PARM       KWD(LIB) TYPE(*NAME) LEN(10) DFT(QGPL) +
          PROMPT('Library')

```

Upcoming 2003 IBM Technical Conferences

Content Management & WebSphere Portal Technical Conference

June 9-11, 2003 *
Orlando, Florida

IBM eServer iSeries Technical Conferences

June 23-27, 2003 *
Las Vegas, Nevada
October 13-17, 2003 *
Orlando, Florida

Storage and Storage Networking Symposium

July 28-August 1, 2003 *
Chicago, Illinois

Networking Solutions Technical Conference

August 25-29, 2003 *
Miami Beach, Florida

SecureWorld: The IBM End-to-End Security Conference

August 25-29, 2003 *
Miami Beach, Florida

IBM eServer xSeries Technical Conference

September 8-12, 2003 *
Orlando, Florida

IMS Technical Conference

September 15-18, 2003 *
Las Vegas, Nevada

IBM eServer pSeries (RS/6000) and Linux Technical University

September 22-26, 2003 *
Miami Beach, Florida

IBM Data Management Technical Conference

October 27-31, 2003 *
Las Vegas, Nevada

Check the IBM Training Web site for more info:

www.ibm.com/training/conf